```
J = N[Exp[2 Pi I / 3], 200]; eta = N[Exp[2 Pi I / 6], 200];

c = {{1 - J, J^2, 0, -2 J}, {J^2, 0, 0, J^2},
   {J^2 - 1, J - 1, J, -2}, {J - 1, J - J^2, 1 - J^2, 2 J}};

alpha = N[Root[1 + 2 #1 + 8 #1^2 + 5 #1^3 + 8 #1^4 + 2 #1^5 + #1^6 &, 2], 100];
beta = N[Root[1 + 2 #1 + 8 #1^2 + 5 #1^3 + 8 #1^4 + 2 #1^5 + #1^6 &, 4], 100];
mu = N[Root[1 + 2 #1 + 8 #1^2 + 5 #1^3 + 8 #1^4 + 2 #1^5 + #1^6 &, 5], 100];

va = {-(-J + 2 J alpha) / ((-1 + alpha) (J + alpha)),
  (eta (1 + J - alpha)) / ((-1 + alpha) (J + alpha)), (eta (1 + J + 2 J alpha)) / (eta + alpha^2), 1};
vb = {-(-J + 2 J beta) / ((-1 + beta) (J + beta)), (eta (1 + J - beta)) / ((-1 + beta) (J + beta)),
  (eta (1 + J + 2 J beta)) / (eta + beta^2), 1};
vm = {-(-J + 2 J mu) / ((-1 + mu) (J + mu)), (eta (1 + J - mu)) / ((-1 + mu) (J + mu)),
  (eta (1 + J + 2 J mu)) / (eta + mu^2), 1}; v1 = {J, 1, 0, 0};

R = Abs[alpha^3]; r = Log[R]; Ta = Arg[alpha^3]; Tb = Arg[beta^3]; Tm = -Ta - Tb;

p = Transpose[{va, v1, vm, vb}];
dx = {{r + I Ta, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, I Tm, 0}, {0, 0, 0, -r + I Tb}};
pdxip = p.dx.Inverse[p];

A = {{0, 0, 1, J^2}, {J, 0, 0, J}, {0, J, 0, J}, {0, 0, 0, -J^2}};

X = Table[Flatten[MatrixPower[A, i].pdxip.MatrixPower[A, -i]], {i, 0, 17}];
x = Table[MatrixPower[A, i].pdxip.MatrixPower[A, -i], {i, 0, 17}];

Distance[m_List, k_Integer] := N[
   Norm[Transpose[Take[m, k]].LeastSquares[Transpose[Take[m, k]], m[[k + 1]]] - m[[k + 1]]]];

Map[Distance[X, #] &, Range[Length[X] - 1]]
```

```
{7.06075, 6.72807, 6.19009, 7.17648, 5.69842,
 4.08814, 2.05316, 1.37461, 0., 0., 0., 0., 0., 0., 0., 0., 0.}
```

```
(* There are 8 nonzero entries,
so we get dimension 9 with this collection of vectors. Now
 let us conjugate the resulting independent vectors with the matrix B. *)
X = Take[X, 9];
x = Take[x, 9];


B = {{0, J^2 - 1, J       , 0},
     {0,    J   , 0      , 0},
     {J, J - J^2, 1 - J^2, 0},
   {1 - J^2, 1 - J^2, 1 - J, 1}};


Y = Flatten[Table[
    Flatten[MatrixPower[B, i].x[[j]].MatrixPower[B, -i]], {i, 1, 5}, {j, Length[x]}], 1];
y = Flatten[Table[MatrixPower[B, i].x[[j]].MatrixPower[B, -i],
    {i, 1, 5}, {j, Length[x]}], 1];
```

```
XY = Join[X, Y];
dist = Map[Distance[XY, #] &, Range[Length[XY] - 1]]
```

```
{7.06075, 6.72807, 6.19009, 7.17648, 5.69842, 4.08814, 2.05316,
 1.37461, 2.61148, 0., 6.20493, 6.92893, 2.55798, 3.68679, 2.81964, 0.,
 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0.,
 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0., 0.}
```

```
(* Note that there is a new entry, than zero,
then 5 nonzero entries. Thus the basic vectors are
  {X1, ..., X9}={X,AXA^{-1},...,A^8 X A^{-8},
  {Y1, Y3, ... Y7}= BXB^{-1}, B A^2 X A^{-2} B^{-1}, ..., B A^6 X A^{-6} B^{-1}  *)
```

```
Select[dist, Positive]
Length[Select[dist, Positive]]
```

```
{7.06075, 6.72807, 6.19009, 7.17648, 5.69842, 4.08814,
 2.05316, 1.37461, 2.61148, 6.20493, 6.92893, 2.55798, 3.68679, 2.81964}
```

```
14
```

```
(* The dimension of our space is at least the latter number plus one,
since we start counting distances from the second
 vector. Thus we have found 15 independent vectors. *)
N[y[[2]]]
```

```
{{-3.23019 - 1.16188 i, -2.62131 + 2.21648 i, -1.88856 + 2.15104 i, -1.1558 + 1.29885 i},
 {-1.34163 - 0.0715233 i, -0.732755 + 1.12612 i, -1.1558 + 0.881874 i, 0. + 1.33461 i},
 {-0.185823 + 2.288 i, 1.88856 + 1.30493 i, 2.07438 + 1.0546 i, 1.70274 + 1.19764 i},
 {3.96294 + 2.288 i, 3.96294 - 2.288 i, 2.07438 - 3.59293 i, 1.88856 - 1.01884 i}}
```

```
N[B.A.pdxip.Inverse[B.A]]
```

```
{{-3.23019 - 1.16188 i, -2.62131 + 2.21648 i, -1.88856 + 2.15104 i, -1.1558 + 1.29885 i},
 {-1.34163 - 0.0715233 i, -0.732755 + 1.12612 i, -1.1558 + 0.881874 i, 0. + 1.33461 i},
 {-0.185823 + 2.288 i, 1.88856 + 1.30493 i, 2.07438 + 1.0546 i, 1.70274 + 1.19764 i},
 {3.96294 + 2.288 i, 3.96294 - 2.288 i, 2.07438 - 3.59293 i, 1.88856 - 1.01884 i}}
```

```
Y = Take[Y, 7];
Y = Delete[Y, 2];
XY = Join[X, Y];
dist = Map[Distance[XY, #] &, Range[Length[XY] - 1]]
```

```
{7.06075, 6.72807, 6.19009, 7.17648, 5.69842, 4.08814,
 2.05316, 1.37461, 2.61148, 6.20493, 6.92893, 2.55798, 3.68679, 2.81964}
```

```
Select[dist, Positive]
Length[Select[dist, Positive]]
```

```
{7.06075, 6.72807, 6.19009, 7.17648, 5.69842, 4.08814,
 2.05316, 1.37461, 2.61148, 6.20493, 6.92893, 2.55798, 3.68679, 2.81964}
```

```
14
```

```
N[Y[[6]]] - Flatten[N[B.MatrixPower[A, 6].pdxip.Inverse[B.MatrixPower[A, 6]]]]
```

```
{0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i,
 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i, 0. + 0. i}
```